
ABSTRACT  

Creating table of contents (TOC) files manually for large RTF report packages is time-consuming and error-prone, 
highlighting the need for a more efficient and reliable approach. After evaluating existing TOC generation methods 
and considering implementation experience, an Excel-based architecture was chosen for its accessibility, flexibility, 
and support for cross-functional collaboration. This paper describes an automated two-step framework that builds 
a navigable Excel-based TOC directly from SAS®-generated RTF reports without requiring predefined metadata. In 
the first step, a SAS or VBA macro recursively scans specified folders and subfolders, parses each RTF file, and extracts 
key information such as file paths, file names, and report titles into a structured Excel workbook. In the second step, 
an automated macro adds hyperlinks to each report entry, allowing reviewers to open individual reports with a single 
click. This framework makes TOC generation more streamlined, scalable, and efficient regardless of report volume, 
and it works in both local file systems and SharePoint environments, providing an enterprise-ready solution for 
report review, document navigation, and cross-functional document management. 

INTRODUCTION  

Tables, listings, and figures (TLFs) are the cornerstone of clinical study reports (CSRs) in the pharmaceutical industry. 
Governed by the ICH E3 guideline, which mandates their inclusion in structured sections of CSRs, TLFs serve as the 
primary mechanism by which statistical programmers and analysts communicate trial efficacy and safety results to 
medical writers, reviewers, regulatory agencies, and quality checkers. A typical Phase 3 study can generate dozens 
to hundreds of individual TLF outputs, sometimes spanning tens of thousands of pages in aggregate. Navigating this 
volume of material — whether for internal review, quality control, or regulatory submission — presents a persistent 
and significant operational challenge. An accurate, comprehensive, and annotatable table of contents (TOC) 
therefore plays a central role in ensuring that all TLF outputs are reviewed thoroughly, and none are overlooked. 

Rich Text Format (RTF) has long been the dominant format for individual TLF production in the industry. SAS's Output 
Delivery System (ODS RTF) provides programmers with powerful, flexible control over table layout, pagination, fonts, 
and styles, and standard macros for RTF generation have been in wide use for decades. The format's underlying 
nature as plain-text ASCII makes it particularly amenable to post-processing: an RTF file can be read into a SAS 
dataset, manipulated at the code level, and reassembled — a property that underlies the RTF code manipulation 
approaches described in this paper. 

Several distinct types of TOCs exist in the clinical trial and broader document management landscape. The eCTD 
(electronic Common Technical Document) TOC is a structured XML-based index of the entire submission package. A 
standalone HTML TOC lists all TLF outputs with hyperlinks to individual files. Spreadsheet-based TOCs, typically 
implemented in Excel, offer a flexible, manually maintained format. Finally, SAS®-generated standalone TOC RTF files 
can be produced programmatically alongside the outputs themselves. Among these, the Excel-based TOC is 
particularly well suited to high-quality review workflows: it requires no specialized tooling to create or maintain, can 
accommodate as much contextual information as needed, and supports practical review management features such 
as color-coding outputs by review status and adding annotations to track progress. 

Manually creating an Excel-based TOC is feasible, but it becomes increasingly inefficient as the number of reports 
grows from dozens to hundreds or more. A more scalable approach is to generate the Excel TOC automatically from 
the metadata associated with the TLF reports using SAS or VBA programming. 

Prior work has approached TOC creation and report navigation from several directions. Walker and Cao (2023) 
presented a macro that automatically creates and synchronizes a shell document and its TOC, and Walker (2023) 
described a complementary framework for assembling and executing SAS macros directly from metadata; both 
assume that standardized shells or metadata are available at the outset. Zeng, Li, and Pan (2011) used Visual Basic 

for Applications (VBA) within Microsoft Word® to assemble SAS output list files and generate a TOC, while Xie (2019) 
developed a SAS utility that concatenates existing RTF tables and figures into a single file with a multi-level bookmark 
hierarchy and a hyperlinked TOC. More recently, Glaß (2024) introduced the iTable and iListing project, which 
provides an interactive interface for navigating clinical study report tables and listings. 

These approaches are highly effective when standardized mock-up shells, predefined metadata, or a fixed report 
inventory are available. In many practical situations, however, none of these inputs exist by the time the outputs are 
delivered. In such cases, TOC creation can still be automated by generating the intermediate metadata directly from 
the finished RTF TLFs—the approach taken in this paper. 

This paper presents an automated method for generating an Excel-based TOC directly from RTF TLFs without 
requiring predefined metadata. The approach integrates seamlessly into the clinical reporting workflow and provides 
an efficient mechanism for organizing and reviewing tables, listings, and figures. The overall workflow can be 
summarized as follows: Raw Clinical Data → SDTM/ADaM Datasets → Table/Listing/Figure Generation → Automated 
TOC Creation for Review → TLF Delivery. 

By generating the TOC directly from SAS®-produced RTF outputs, the method streamlines the review process, 
reduces manual effort, and enhances the accessibility and management of large collections of clinical study reports. 

SAS®-GENERATED RTF REPORTS 

SAS®-generated RTF (Rich Text Format) files are text-based documents that combine data output with formatting 
instructions, allowing reports to be viewed and edited in word-processing applications such as Microsoft Word®. 
Because RTF is a widely supported document format, it is commonly used for sharing analytical results, regulatory 
reports, and business documents with users who may not have access to SAS. SAS®-generated RTF files can also 
support features such as page numbering, style templates, and customized layouts, making them suitable for 
professional reporting requirements. 

In SAS, RTF files are typically created using the Output Delivery System (ODS), which provides a flexible framework 
for directing procedure output to various formats. The ODS RTF destination captures the results of SAS procedures 
and writes them to an RTF document while applying a specified style template. Users can control the appearance of 
the report through ODS options, including font selection, colors, margins, and page orientation. 

The following SAS code examples illustrate the generation of RTF output for tables, listings, and figures (TLFs) using 
a customized style template. 

*** Customize RTF style template.; 

proc template; 

  define style RTFTemplate; 

  *** Include style details here.; 

end; 

run; 

 

*** Open the RTF destination.; 

ods rtf body="&tlf_folder./&tlf_filename..rtf" style=RTFTemplate; 

ods escapechar='^'; 

 

*** Assign titles for current report.; 

title1 j=c "&tlf_number."; 

title2 j=c "&tlf_main_title."; 

title3 j=c "&tlf_population."; 

 

*** Include any SAS code here, including the procedures used to generate 

the main content of the table, listing, or figure.;  

 

*** Close the RTF destination.; 

ods rtf close; 

In the code above, the strings after & are the pre-defined macro variables. An escape character (^n) can be added 
to the title text (defined in &tlf_main_title or &tlf_population) to insert a line break. Once all tables, listings, and 
figures (TLFs) have been generated using a common reporting style template, the title information—including the 
TLF number, main title, and population, which comprise the key information displayed in the table of contents—can 
be extracted directly from the RTF output files using SAS or VBA programs. The extraction process and 
implementation details are described in the following section. 

If the titles in the RTF outputs are defined using statements other than TITLEn, or if the RTF outputs are generated 
without using the ODS RTF destination, adding invisible special text strings (which are not displayed when the RTF 
file is opened in Microsoft Word®) immediately before or after the titles can facilitate the extraction of title 
information later. In an RTF file, you can add a hidden text string by inserting the RTF hidden-text control word \v 
before the text and \v0 after the text. The \v control word makes subsequent characters hidden, and \v0 turns 
hidden formatting off. For example, the following two lines in an RTF file produce the same visible text when opened 
in Microsoft Word®. However, in the second example, the title can be easily identified by searching for the hidden 
string TLF main title using SAS or VBA programs. 

\fs16 Listing of Demographics 

\fs16 Listing of Demographics {\v TLF main title\v0} 

SCAN RTF REPORTS FOR TITLES  

Once the RTF outputs are available, the first step of the framework is to scan the delivery folder tree, locate every 
RTF file, and recover the identifying title information from each one. Two independent implementations are 
provided—one in SAS and one in VBA—so the step can be run wherever it is most convenient. SAS is well suited to 
the server or SAS® Life Science Analytics Framework (LSAF) environment where the outputs are produced, while VBA 
runs entirely inside Microsoft Excel® on a reviewer’s desktop with no SAS session required. Both implementations 
follow the same two-stage logic: (1) recursively list all RTF files beneath a root folder, and (2) parse each file to extract 
its report number and titles. 

SAS IMPLEMENTATION 

The %list_all_files macro performs a pure-SAS recursive directory walk using the DOPEN, DNUM, DREAD, and FINFO 
functions, so it needs no X commands or pipe access and runs unchanged on Windows, UNIX, and LSAF. Each RTF 
file found is written as one observation to an output dataset, capturing its full path, file name, extension, folder, 
depth, size, and last-modified date. The core of the recursion is shown below. 

%let did    = %sysfunc(dopen(&dirref.));  *** open the directory ***; 

%let memcnt = %sysfunc(dnum(&did.));       *** number of members ***; 

%do i = 1 %to &memcnt.; 

  %let memname = %qsysfunc(dread(&did., &i.)); 

  *** recurse into a subfolder, or record the RTF file with FINFO ***; 

%end; 

 

Each RTF file is then parsed by the %ttls_extract macro. The macro reads the raw RTF text, keeps only the block 
between the {\header and {\footer control words (where the TITLEn statements are rendered), splits that block into 
rows on the {\row} marker, and pulls the visible text out of each {…\cell}. All RTF control words are removed with a 
single PRXCHANGE regular-expression call (Borowiak 2013), the braces are dropped, and runs of blanks are 
collapsed. 

*** Retain only the text between the header and footer keywords, ***; 

*** and then combine all lines into a single record. ***; 

retain text; 

if line=:'{\header' then text = strip(line); 

if line=:'{\footer' then do; output; stop; end; 

text = catx(' ', text, strip(line)); 

*** Identify all row titles from the above retained text. ***;  

pos = 1; rownum = 0; 

** Transform line breaks into " ~ " separators **; 

text = tranwrd(text, "{\line}", " ~ "); 

do while (1); 

  ** Find next {\row} **; 

  row_end = index(substr(text,pos),'{\row}'); 

  if row_end = 0 then leave; 

  row = substr(text,pos,row_end-1); 

  rownum + 1; rowtitle = ''; 

  remain = row; 

  ** Extract every cell in this row **; 

  do while (index(remain,'\cell}')); 

    cell_end = index(remain,'\cell}'); 

    ** Locate the first { in the remain text **; 

    brace = index(remain, '{'); 

    if brace then do; 

      cell = substr(remain,brace+1,cell_end-brace-1); 

      ** Remove all RTF tags **; 

      cell = prxchange('s/\\[A-Za-z*]+\d*//',-1,cell); 

      ** Remove braces **; 

      cell = compress(cell,'{}'); 

      ** Compress blanks **; 

      cell = compbl(strip(cell)); 

      ** Connect multiple cells within each row with | **; 

      rowtitle = catx(' | ',rowtitle,cell); 

    end; 

    remain = substr(remain,cell_end+5); 

  end; 

  output; 

  pos = pos + row_end + 5; 

end; 

keep rownum rowtitle; 

 

**** Transform all titles as one row per output RTF file ****; 

proc transpose data=title_rows out=title_cols prefix=title; 

  id rownum; var rowtitle; 

run; 

 

A driver step loops over every RTF file, calls %ttls_extract, and stacks the results, so that each RTF output becomes 
one observation in the dataset title_cols whose variables title1, title2, title3, … correspond to successive header 
rows. 

%do %while (&thisrtf. < &allrtf.); 

  %let thisrtf = %eval(&thisrtf. + 1); 

  data _null_; set _all_rtf; where order = &thisrtf.; 

    call symput('rtfpath', strip(filepath)); 

    call symput('rtfname', strip(filename)); 

  run; 

  %ttls_extract(rtffile=&rtfpath.); *** extract titles from one RTF ***; 

  data alltitles; set alltitles title_cols; run; 

%end; 

VBA IMPLEMENTATION 

The VBA implementation in Microsoft Excel® reproduces the same logic using the Scripting.FileSystemObject. The 
ListRtfFiles macro reads a root folder from cell B2 (interpreted as relative to the workbook, or as an absolute path), 
walks the folder tree recursively, and writes one row per RTF file—its subfolder and file name—to a temporary 
worksheet named list_of_rtfs. 

' Recursively scan a folder, writing one row per .rtf file 

Private Sub ScanFolder(fso, folder, baseFolder, ws, nextRow) 

  For Each f In folder.Files 

    If LCase$(fso.GetExtensionName(f.Name)) = "rtf" Then 

      ws.Cells(nextRow, 1).Value = RelativeSubfolder(folder.Path, 

baseFolder) 

      ws.Cells(nextRow, 2).Value = f.Name 

      nextRow = nextRow + 1 

    End If 

  Next f 

  For Each subf In folder.SubFolders        ' recurse into subfolders 

    ScanFolder fso, subf, baseFolder, ws, nextRow 

  Next subf 

End Sub 

 

The ExtractRtfTitles macro then opens each listed RTF and applies the same header-to-footer, row-by-row, cell-by-
cell parsing used in SAS. A pre-compiled VBScript regular expression (pattern \\[A-Za-z\*]+-?\d*) strips the RTF 
control words, and the cleaned cells of each row are joined with “ | ” and written back to the sheet as columns Title1, 
Title2, Title3, … 

' Clean one RTF cell to plain text (mirrors the SAS ttls_extract logic) 

Private Function CleanCell(rx, raw) As String 

  Dim s As String: s = raw 

  s = Replace(s, "{\line}", " ~ ")   ' in-cell line breaks -> ~ 

  s = rx.Replace(s, "")               ' drop RTF control words 

  s = Replace(s, "{", ""): s = Replace(s, "}", "") 

  CleanCell = CompressBlanks(s)        ' collapse blanks + trim 

End Function 

 

GENERATE EXCEL TOC WITH HYPERLINKS  

After the titles have been extracted, the second step derives the table-of-contents columns from the title rows and 
writes a navigable Excel workbook in which each report’s file name is a clickable hyperlink. Depending on the 
reporting template, the RTF header may also contain common boilerplate lines—for example, a sponsor name or a 
confidentiality statement—which are extracted as additional titles. These extra entries can be dropped, and the 
retained text can be relabeled as needed to produce more informative TOC column headers. In this paper, the TOC 
uses six columns: Type, Number, Title, Population, Output, and Comments. The Comments column is left blank for 
reviewers to record review status or notes. 

SAS IMPLEMENTATION 

In SAS, the stacked titles dataset is post-processed to split title1 into Type (the first word) and Number (the 
remainder), and to map title2 to Title and title3 to Population. The Output column is built as an Excel HYPERLINK() 
formula that points to the corresponding RTF file, and PROC PRINT is directed to an ODS EXCEL destination with an 
auto-filter enabled on the header row. 

*** Derive the TOC columns from the extracted titles ***; 

Type       = scan(title1, 1, ' '); 

Number     = scan(title1, 2, ' '); 

Title      = strip(title2); 

Population = strip(title3); 

*** Build a clickable hyperlink to each RTF output ***; 

Output = '=HYPERLINK("' || strip(_output) || '", "' || 

  strip(_output) || '")'; 

*** Create a Comments column for reviewer convenience ***;  

Comments = ' '; 

 

ods excel file="&output_path./toc_rtfs.xlsx" 

  options(sheet_name="TOC" autofilter="1-6"); 

 

proc print data=toc noobs label; 

  var type number title population output comments; 

run; 

 

ods excel close; 

 

VBA IMPLEMENTATION 

In VBA, the CreateTOC macro reads the list_of_rtfs sheet, splits Title1 into Type and Number, and writes a new 
workbook (named in cell C2) containing a single TOC sheet with the six columns, an auto-filter, and fitted column 
widths. Because the workbook is saved first, each Output cell can then be given a relative hyperlink built with forward 
slashes, so the links resolve both from Windows Explorer and from a SharePoint document library. 

' Type = first token of Title1, Number = the remainder 

sp  = InStr(t1, " ") 

typ = Left$(t1, sp - 1): num = Trim$(Mid$(t1, sp + 1)) 

 

' Relative path with forward slashes -> works locally and on SharePoint 

rel = Replace(subFolder, "\", "/") & "/" & fileName 

 

' Add a relative hyperlink on the Output cell after the workbook is saved 

toc.Hyperlinks.Add Anchor:=toc.Cells(hlRows(i), 5), _ 

  Address:=relPaths(i), TextToDisplay:=CStr(toc.Cells(hlRows(i), 5).Value) 

 

Display 1 shows the worksheet interface used to drive the VBA method. The reviewer enters the delivery folder and 
the desired output file name in the highlighted input cells, then runs the three macros—ListRtfFiles, ExtractRtfTitles, 
and CreateTOC—in sequence using the buttons provided. 

 

Display 1. Interface for Excel TOC Generation via the VBA Method  

DISCUSSION  

Output 1 shows a sample Excel TOC generated using the VBA method. The Output column is displayed in a different 
color before and after its hyperlink is clicked, and the Comments column can be used to record each report’s review 
status. The Excel TOC produced by the SAS method is essentially identical. 

 

Output 2. Sample Excel TOC Generated Using the VBA Method 

 

SAS VERSUS VBA IMPLEMENTATION 

Both implementations produce an equivalent Excel TOC, so the choice between them is largely a matter of where 
the work is most convenient rather than of capability. The SAS implementation is the natural choice when the TOC 
is produced as part of the programming workflow that generates the outputs: it runs on the same server or LSAF 
environment where the RTF files already reside, can be scheduled or called from a driver program, and integrates 
cleanly with existing macro libraries. The VBA implementation is the natural choice for reviewers and cross-
functional colleagues who work primarily in Excel and may not have a SAS license; a user points cell B2 at the delivery 
folder and clicks three macro buttons to list the files, extract the titles, and build the TOC. The two approaches are 
therefore complementary, as summarized in Table 1. 

Consideration SAS Implementation VBA Implementation 

Runtime environment Server or LSAF where outputs are produced Reviewer desktop, inside Excel 

Software required SAS session Microsoft Excel only (no SAS license) 

Directory scan DOPEN / DNUM / DREAD / FINFO recursion FileSystemObject recursion 

Title parsing PRXCHANGE regular expression VBScript RegExp object 

TOC output ODS EXCEL with HYPERLINK() formula Native workbook with Hyperlinks.Add 

Hyperlink style Formula-based link Relative link (local + SharePoint) 

Best suited for Automated production pipeline Ad hoc, cross-functional review 

Table 1. Comparison of the SAS and VBA implementations 

HANDLING NON-STANDARD OUTPUTS 

The framework assumes that all outputs share a common reporting template so that the type, number, main title, 
and population appear in predictable header rows. When outputs are generated outside the ODS RTF destination, 
or when TITLEn statements are not used to define titles, the hidden-text technique described earlier—wrapping a 
marker string between the \v and \v0 control words immediately before or after each title—lets the same parsers 
locate the titles reliably by searching for the invisible marker rather than by row position. This keeps the approach 
adaptable to heterogeneous report packages while still requiring no predefined mock-up documents or external 
metadata. 

In summary, key advantages of the approach in this paper include: 

• No predefined mock-up documents or metadata are required  

• Adapts to different folder structures and report sets  

• Reduces manual effort and the transcription errors it can introduce 

• Makes it easy to track the review status of each report 

 

CONCLUSION 

This paper presents an automated Excel-based framework for generating a navigable Table of Contents from SAS®-
produced RTF reports without predefined metadata. The two-step process—RTF scanning and hyperlink 
generation—significantly improves report accessibility and reduces manual effort in document indexing. The 
proposed approach offers a scalable, efficient, and easily integrated solution for managing large volumes of RTF 
outputs, making it particularly well suited for the production and review of high-quality Tables, Listings, and Figures 
(TLFs) in clinical trial regulatory submissions. 

